# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Contributor: Sergej Pupykin <pupykin.s+aur@gmail.com>

pkgname=perl-patchreader
pkgver=0.9.6
pkgrel=5
pkgdesc="Utilities to read and manipulate patches and CVS"
arch=('any')
url="https://search.cpan.org/dist/PatchReader"
license=('GPL' 'PerlArtistic')
depends=('perl')
options=('!emptydirs')
source=(https://search.cpan.org/CPAN/authors/id/T/TM/TMANNERM/PatchReader-$pkgver.tar.gz)
md5sums=('bd8da2388cd5ebc99a860b6d6bfeb8ad')

build() {
  cd  "$srcdir"/PatchReader-$pkgver
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  cd  "$srcdir"/PatchReader-$pkgver
  make install DESTDIR="$pkgdir"
  find "$pkgdir" -name '.packlist' -delete
  find "$pkgdir" -name '*.pod' -delete
}
